// File: contracts/Nafta.sol



//SPDX-License-Identifier: Unlicense
pragma solidity ^0.8.9;

import "@openzeppelin/contracts/token/ERC721/IERC721.sol";
import "@openzeppelin/contracts/token/ERC721/ERC721.sol";
import "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import "./interfaces/INafta.sol";
import "./interfaces/IFlashNFTReceiver.sol";
import {ERC721Holder} from "@openzeppelin/contracts/token/ERC721/utils/ERC721Holder.sol";
import {Ownable} from "@openzeppelin/contracts/access/Ownable.sol";

contract Nafta is INafta, ERC721, ERC721Holder, Ownable {
  // WETH9 address
  IERC20 immutable WETH9;

  /// @dev mapping from token contract to token id to token details/vault
  mapping(address => mapping(uint256 => PoolNFT)) internal _poolNFTs;

  /// @dev mapping of earnings of each user of the pool (in WETH9)
  mapping(address => uint256) public earnings;

  /// @dev fee taken by pool on each flashloan or flashlong operation
  uint256 public poolFee;
  uint256 public poolFeeChangedAtBlock;

  // Newly proposed owner, who will be able to claim the ownership
  address public proposedOwner;

  // We use a single ERC721 NaftaNFT for both Borrowers and Lenders NFT types.
  // Thus we introduce an ID shift for LenderNFTs type, to distinguish their IDs from BorrowerNFTs
  // So all IDs below 2**32 are BorrowerNFTs, and all above 2**32 are LenderNFTs

  /// @dev keeps track of the next free BorrowerNFT ID (longrent)
  /// @dev is in range 1 to 4 294 967 295 and the ID is used as is in NaftaNFT
  /// @dev 0 is reserved for N/A
  uint256 public borrowerNFTCount = 0;

  /// @dev keeps track of the next free LenderNFT ID (when adding to pool)
  /// @dev is in range 4 294 967 297 to 8 589 934 591 (subtract 2**32 to get actual count)
  /// @dev 4 294 967 296 is reserved for N/A
  uint256 public lenderNFTCount = 2**32;

  constructor(address owner_, IERC20 WETH9_) ERC721("NaftaNFT", "NAFTA") {
    WETH9 = WETH9_;
    Ownable.transferOwnership(owner_);
  }

  /////////////
  // Getters //
  /////////////

  /// @dev poolNFT storage view function
  /// @param nftAddress Address of the NFT contract
  /// @param nftId NFT ID
  function poolNFTs(address nftAddress, uint256 nftId) external view override returns (PoolNFT memory nft) {
    nft = _poolNFTs[nftAddress][nftId];
  }

  ////////////////////
  // Pool functions //
  ////////////////////

  /// @notice Add your NFT to the pool
  /// @param nftAddress - The address of NFT contract
  /// @param nftId - ID of the NFT token you want to add
  /// @param flashFee - The fee user has to pay for a single rent (in WETH9) [Range: 0-4722.36648 ETH] (0 if flashrent is free)
  /// @param pricePerBlock - If renting longterm - this is the price per block (0 if not allowing renting longterm) [Range: 0-4722.36648 ETH]
  /// @param maxLongtermBlocks - Maximum amount of blocks for longterm rent [Range: 0-16777216]
  function addNFT(
    address nftAddress,
    uint256 nftId,
    uint256 flashFee,
    uint256 pricePerBlock,
    uint256 maxLongtermBlocks
  ) external {
    // Protection from weird and meaningless adding of NaftaNFTs
    require(nftAddress != address(this), "Can't add Nafta NFTs to Nafta");
    // Verify that NFT isn't already in the pool
    require(_poolNFTs[nftAddress][nftId].lenderNFTId == 0, "NFT is already in the Pool");

    // In Longterm rent, we have protection from cheaters who want to rent longterm for 1 block and pay less than a flash loan fee.
    // This check is to make sure if longrent with these parameters is even possible.
    // If pricePerBlock == 0 then Longterm rent is disabled.
    require((pricePerBlock == 0) || (maxLongtermBlocks * pricePerBlock >= flashFee), "Max Longterm rent can't be cheaper than flashloan");

    // Pull the NFT from the msg.sender using transferFrom
    IERC721(nftAddress).safeTransferFrom(msg.sender, address(this), nftId);

    uint256 newNFTId = lenderNFTCount + 1;
    lenderNFTCount = newNFTId;

    // Store newly added NFT renting parameters after packing to struct
    _poolNFTs[nftAddress][nftId] = fromBigPoolNFT(
      // (flashFee, pricePerBlock, maxLongtermBlocks, inLongtermTillBlock, borrowerNFTId, lenderNFTId)
      BigPoolNFT(flashFee, pricePerBlock, maxLongtermBlocks, 0, 0, newNFTId)
    );

    // Mint a new LenderNFT to msg.sender
    _safeMint(msg.sender, newNFTId);

    // Emit AddNFT event
    emit AddNFT(nftAddress, nftId, flashFee, pricePerBlock, maxLongtermBlocks, newNFTId, msg.sender);
  }

  /// @notice Edit your NFT prices
  /// @param nftAddress - The address of NFT contract
  /// @param nftId - ID of the NFT token you have in the pool
  /// @param flashFee - The fee user has to pay for a single rent (in WETH9) [Range: 0-4722.36648 ETH] (0 if flashrent is free)
  /// @param pricePerBlock - If renting longterm - this is the price per block (0 if not allowing renting longterm) [Range: 0-4722.36648 ETH]
  /// @param maxLongtermBlocks - Maximum amount of blocks for longterm rent [Range: 0-16777216]
  function editNFT(
    address nftAddress,
    uint256 nftId,
    uint256 flashFee,
    uint256 pricePerBlock,
    uint256 maxLongtermBlocks
  ) external {
    // In Longterm rent, we have protection from cheaters who want to rent longterm for 1 block and pay less than a flash loan fee.
    // This check is to make sure if longrent with these parameters is even possible.
    // If pricePerBlock == 0 then Longterm rent is disabled.
    require((pricePerBlock == 0) || (maxLongtermBlocks * pricePerBlock >= flashFee), "Max Longterm rent can't be cheaper than flashloan");

    BigPoolNFT memory bigPoolNFT = toBigPoolNFT(_poolNFTs[nftAddress][nftId]);

    // Verify that msg.sender is stored as an owner of this NFT
    require(ownerOf(bigPoolNFT.lenderNFTId) == msg.sender, "Only owner of the corresponding LenderNFT can call this");

    // Update parameters: flashFee, pricePerBlock, maxLongtermBlocks
    bigPoolNFT.flashFee = flashFee;
    bigPoolNFT.pricePerBlock = pricePerBlock;
    bigPoolNFT.maxLongtermBlocks = maxLongtermBlocks;

    // Save the updated NFT back to storage after packing
    _poolNFTs[nftAddress][nftId] = fromBigPoolNFT(bigPoolNFT);

    // Emit EditNFT event
    emit EditNFT(nftAddress, nftId, flashFee, pricePerBlock, maxLongtermBlocks, bigPoolNFT.lenderNFTId, msg.sender);
  }

  /// @notice Remove your NFT from the pool with earnings
  /// @param nftAddress - The address of NFT contract
  /// @param nftId - ID of the NFT token you want to remove
  function removeNFT(address nftAddress, uint256 nftId) external {
    BigPoolNFT memory bigPoolNFT = toBigPoolNFT(_poolNFTs[nftAddress][nftId]);

    // Verify that msg.sender is stored as an owner of this NFT
    require(ownerOf(bigPoolNFT.lenderNFTId) == msg.sender, "Only owner of the corresponding LenderNFT can call this");
    // Verify that it's not rented in longterm right now
    require(bigPoolNFT.inLongtermTillBlock < block.number, "Can't remove NFT from the pool while in longterm rent");
    // If it's not rented longterm - update longterm rent and burn if needed
    updateLongRent(nftAddress, nftId);

    // Zero the storage of this NFT in the pool and burn the NFT
    delete _poolNFTs[nftAddress][nftId];
    _burn(bigPoolNFT.lenderNFTId);

    // Push the NFT to msg.sender
    IERC721(nftAddress).safeTransferFrom(address(this), msg.sender, nftId);

    // Emit RemoveNFT event
    emit RemoveNFT(nftAddress, nftId, bigPoolNFT.lenderNFTId, msg.sender);
  }

  /// @notice Withdraw the earnings of your NFT
  function withdrawEarnings() external override {
    uint256 transferAmount = earnings[msg.sender];

    // Verify that earnings > 0
    require(transferAmount > 0, "No earnings to withdraw");

    // Reset earnings before transfer
    earnings[msg.sender] = 0;

    // Push the WETH9 earnings to msg.sender
    require(WETH9.transfer(msg.sender, transferAmount), "WETH9 transfer failed");

    // Emit withdraw event
    emit WithdrawEarnings(transferAmount, msg.sender);
  }

  /// @notice Execute a Flashloan of NFT
  /// @param nftAddress - The address of NFT contract
  /// @param nftId - ID of the NFT token you want to flashloan
  /// @param maxLoanPrice - Price the user is willing to pay for the flashloan (protection from editing frontrunning)
  /// @param receiverAddress - the contract that will receive the NFT (has to implement INFTFlashLoanReceiver interface)
  /// @param data - calldata that will be passed to the receiver contract (optional)
  function flashloan(
    address nftAddress,
    uint256 nftId,
    uint256 maxLoanPrice,
    address receiverAddress,
    bytes calldata data
  ) external {
    // Verify that this NFT still exists in the pool
    require(IERC721(nftAddress).ownerOf(nftId) == address(this), "NFT should be in the pool");

    // Update longterm rent parameters
    updateLongRent(nftAddress, nftId);

    BigPoolNFT memory bigPoolNFT = toBigPoolNFT(_poolNFTs[nftAddress][nftId]);

    // Is this NFT already in a longterm rent?
    bool longterm = bigPoolNFT.inLongtermTillBlock >= block.number;

    // If this NFT is in a longterm rent right now - check if the msg.sender has the BorrowerNFT
    if (longterm) {
      require(
        ownerOf(bigPoolNFT.borrowerNFTId) == msg.sender,
        "This NFT is in longterm rent - you can't flashloan it unless you have corresponding BorrowerNFT"
      );
    }

    uint256 lenderFees = longterm ? 0 : bigPoolNFT.flashFee;

    require(lenderFees <= maxLoanPrice, "You can't take the flashloan for the indicated price");

    // Initialize the Receiver with IFlashNFTReceiver
    IFlashNFTReceiver receiver = IFlashNFTReceiver(receiverAddress);

    // Push the NFT to Receiver
    IERC721(nftAddress).safeTransferFrom(address(this), receiverAddress, nftId);

    require(receiver.executeOperation(nftAddress, nftId, lenderFees, msg.sender, data), "Error during FlashNFT Execution");

    // Pull the NFT back from Receiver (will revert if not possible)
    IERC721(nftAddress).safeTransferFrom(receiverAddress, address(this), nftId);

    // Pull the flashFee fee from msg.sender (will revert if not possible)
    require(longterm || WETH9.transferFrom(msg.sender, address(this), lenderFees), "Can't transfer WETH9 lender fees");

    // Calculate the part of fee that goes to the pool (flashFee * poolFee)
    uint256 poolPart = (poolFee * lenderFees) / 1e18;

    // Add poolPart to the pool Owner balance if it's more than zero
    if (poolPart > 0) earnings[owner()] += poolPart;

    // Add the received (flashFee - poolPart) to NFT owner's earnings
    earnings[ownerOf(bigPoolNFT.lenderNFTId)] += lenderFees - poolPart;

    // This might be an excess check, because we had a successful "safeTransferFrom" above,
    // but better be safe than sorry.
    require(IERC721(nftAddress).ownerOf(nftId) == address(this), "NFT should be in the pool");

    // Emit a Flashloan event
    emit Flashloan(nftAddress, nftId, lenderFees, msg.sender);
  }

  ///////////////////
  // Longterm rent //
  ///////////////////

  /// @dev Utility function to update the status of longterm rent
  /// @dev Is called in some functions as well as can be called by public
  /// @dev will reset the values and burn the BorrowerNFT if the longterm rent is over
  /// @param nftAddress - The address of NFT contract
  /// @param nftId - ID of the NFT token you want to update
  function updateLongRent(address nftAddress, uint256 nftId) public {
    BigPoolNFT memory bigPoolNFT = toBigPoolNFT(_poolNFTs[nftAddress][nftId]);

    if (bigPoolNFT.inLongtermTillBlock > 0 && bigPoolNFT.inLongtermTillBlock < block.number) {
      _burn(bigPoolNFT.borrowerNFTId);

      bigPoolNFT.borrowerNFTId = 0;
      bigPoolNFT.inLongtermTillBlock = 0;

      _poolNFTs[nftAddress][nftId] = fromBigPoolNFT(bigPoolNFT);
    }
  }

  /// @notice Take the NFT into longterm rent
  /// @dev You can take an NFT into longterm rent - meaning you won't have to pay fees for each flashloan use.
  /// @dev You also get exclusivity - nobody else will be able to use it either while your rent lasts (even the original owner!)
  /// @param nftAddress - The address of NFT contract
  /// @param nftId - ID of the NFT token you want to rent
  /// @param maxPricePerBlock - Price the user is willing to pay per block for renting the NFT (protection from editing frontrunning)
  /// @param receiverAddress - Who will receive the longterm rent BorrowerNFT
  /// @param blocks - How many blocks you want to rent (price is calculated per-block)
  function longRent(
    address nftAddress,
    uint256 nftId,
    uint256 maxPricePerBlock,
    address receiverAddress,
    uint256 blocks
  ) external {
    BigPoolNFT memory bigPoolNFT = toBigPoolNFT(_poolNFTs[nftAddress][nftId]);

    // If pricePerBlock is 0 - longterm renting is disabled for this NFT
    require(bigPoolNFT.pricePerBlock > 0, "This NFT isn't available for longterm rent");

    require(blocks <= bigPoolNFT.maxLongtermBlocks, "NFT can't be rented for that amount of time");
    // We don't check for (blocks > 0) because we check it with (longtermPayment >= bigPoolNFT.flashFee) below

    updateLongRent(nftAddress, nftId);

    // Check if it's not in longterm rent already
    require(bigPoolNFT.inLongtermTillBlock < block.number, "Can't rent longterm because it's already rented");

    require(bigPoolNFT.pricePerBlock <= maxPricePerBlock, "Can't rent the NFT with the selected price");

    // Set a new blocknumber for longterm rent
    // This shouldn't overflow because we check (blocks <= maxLongtermBlocks) which is 24bit (16 777 215),
    // and inLongtermTillBlock is 32 bit (4 294 967 295).
    // So theoretically this will be allright until block 4278190080 - which is millenias ahead.
    bigPoolNFT.inLongtermTillBlock = block.number + blocks;

    uint256 longtermPayment = blocks * bigPoolNFT.pricePerBlock;

    // Protecting from cheaters who want to rent longterm for 1 block and pay less than a flash loan fee
    require(longtermPayment >= bigPoolNFT.flashFee, "Longterm rent can't be cheaper than flashloan");

    // Pull the money from lender
    require(WETH9.transferFrom(msg.sender, address(this), longtermPayment), "Can't transfer WETH9 lender fees");

    // Calculate the part of fee that goes to the pool (flashFee * poolFee)
    uint256 poolPart = (poolFee * longtermPayment) / 1e18;

    // Add poolPart to the pool Owner balance if it's more than zero
    if (poolPart > 0) earnings[owner()] += poolPart;

    // Add the rest (flashFee - calculatedPoolFee) to NFT owner's earnings
    earnings[ownerOf(bigPoolNFT.lenderNFTId)] += longtermPayment - poolPart;

    uint256 newNFTId = borrowerNFTCount + 1;
    borrowerNFTCount = newNFTId;

    bigPoolNFT.borrowerNFTId = newNFTId;
    _poolNFTs[nftAddress][nftId] = fromBigPoolNFT(bigPoolNFT);

    _safeMint(receiverAddress, newNFTId);

    emit LongtermRent(nftAddress, nftId, blocks, longtermPayment, newNFTId, msg.sender);
  }

  /////////////////////
  // Admin Functions //
  /////////////////////

  /// @dev Change the pool fee (admin only)
  /// @dev Only admin should be able to do that
  /// @dev Only once per block and for one percentage point - to prevent frontrunning
  /// @param newPoolFee - The new pool fee value (percentage, where 100% is 1e18)
  function changePoolFee(uint256 newPoolFee) external onlyOwner {
    uint256 diff = newPoolFee > poolFee ? (newPoolFee - poolFee) : (poolFee - newPoolFee);
    require(diff <= 1e16, "Can't change the pool fee more than one percentage point in one step");
    require(block.number != poolFeeChangedAtBlock, "Can't change the pool fee more than once in a block");
    poolFee = newPoolFee;
    poolFeeChangedAtBlock = block.number;
    emit PoolFeeChanged(newPoolFee);
  }

  /// @dev Propose a new owner, who will be able to claim ownership over this contract
  /// @param newOwner New owner who will be able to claim ownership over this contract
  function proposeNewOwner(address newOwner) external onlyOwner {
    proposedOwner = newOwner;
  }

  /// @dev Claims the ownership of the contract if msg.sender is proposedOwner
  function claimOwnership() external {
    require(msg.sender == proposedOwner, "Only proposed owner can claim the ownership");
    Ownable._transferOwnership(msg.sender);
  }

  ///////////////////////
  // Utility Functions //
  ///////////////////////

  /// @dev Converts packed PoolNFT struct to unpacked BigPoolNFT struct (uint256)
  /// @param poolNFT packed PoolNFT struct
  /// @return unpacked BigPoolNFT struct
  function toBigPoolNFT(PoolNFT memory poolNFT) public pure returns (BigPoolNFT memory) {
    return
      BigPoolNFT(
        uint256(poolNFT.flashFee),
        uint256(poolNFT.pricePerBlock),
        uint256(poolNFT.maxLongtermBlocks),
        uint256(poolNFT.inLongtermTillBlock),
        uint256(poolNFT.borrowerNFTId),
        uint256(poolNFT.lenderNFTId) + 2**32
      );
  }

  /// @dev Converts unpacked BigPoolNFT struct (uint256) to packed PoolNFT struct
  /// @param bigPoolNFT unpacked BigPoolNFT struct
  /// @return packed PoolNFT struct
  function fromBigPoolNFT(BigPoolNFT memory bigPoolNFT) public pure returns (PoolNFT memory) {
    // Check for overflows before downcasting (yes, Solidity 0.8 still doesn't revert during downcast!)
    require(bigPoolNFT.flashFee <= type(uint72).max, "flashFee doesn't fit in uint72");
    require(bigPoolNFT.pricePerBlock <= type(uint72).max, "pricePerBlock doesn't fit in uint72");
    require(bigPoolNFT.maxLongtermBlocks <= type(uint24).max, "maxLongtermBlocks doesn't fit in uint24");
    require(bigPoolNFT.inLongtermTillBlock <= type(uint32).max, "inLongtermTillBlock doesn't fit in uint32");
    require(bigPoolNFT.borrowerNFTId <= type(uint32).max, "borrowerNFTId doesn't fit in uint32");
    require(bigPoolNFT.lenderNFTId - 2**32 <= type(uint32).max, "lenderNFTId doesn't fit in uint32");

    return
      PoolNFT(
        uint72(bigPoolNFT.flashFee),
        uint72(bigPoolNFT.pricePerBlock),
        uint24(bigPoolNFT.maxLongtermBlocks),
        uint32(bigPoolNFT.inLongtermTillBlock),
        uint32(bigPoolNFT.borrowerNFTId),
        uint32(bigPoolNFT.lenderNFTId - 2**32)
      );
  }
}

// File: @openzeppelin/contracts/token/ERC721/IERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0-rc.1 (token/ERC721/IERC721.sol)

pragma solidity ^0.8.0;

import "../../utils/introspection/IERC165.sol";

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {safeTransferFrom} whenever possible.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool _approved) external;

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes calldata data
    ) external;
}

// File: @openzeppelin/contracts/token/ERC721/ERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0-rc.1 (token/ERC721/ERC721.sol)

pragma solidity ^0.8.0;

import "./IERC721.sol";
import "./IERC721Receiver.sol";
import "./extensions/IERC721Metadata.sol";
import "../../utils/Address.sol";
import "../../utils/Context.sol";
import "../../utils/Strings.sol";
import "../../utils/introspection/ERC165.sol";

/**
 * @dev Implementation of https://eips.ethereum.org/EIPS/eip-721[ERC721] Non-Fungible Token Standard, including
 * the Metadata extension, but not including the Enumerable extension, which is available separately as
 * {ERC721Enumerable}.
 */
contract ERC721 is Context, ERC165, IERC721, IERC721Metadata {
    using Address for address;
    using Strings for uint256;

    // Token name
    string private _name;

    // Token symbol
    string private _symbol;

    // Mapping from token ID to owner address
    mapping(uint256 => address) private _owners;

    // Mapping owner address to token count
    mapping(address => uint256) private _balances;

    // Mapping from token ID to approved address
    mapping(uint256 => address) private _tokenApprovals;

    // Mapping from owner to operator approvals
    mapping(address => mapping(address => bool)) private _operatorApprovals;

    /**
     * @dev Initializes the contract by setting a `name` and a `symbol` to the token collection.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(ERC165, IERC165) returns (bool) {
        return
            interfaceId == type(IERC721).interfaceId ||
            interfaceId == type(IERC721Metadata).interfaceId ||
            super.supportsInterface(interfaceId);
    }

    /**
     * @dev See {IERC721-balanceOf}.
     */
    function balanceOf(address owner) public view virtual override returns (uint256) {
        require(owner != address(0), "ERC721: balance query for the zero address");
        return _balances[owner];
    }

    /**
     * @dev See {IERC721-ownerOf}.
     */
    function ownerOf(uint256 tokenId) public view virtual override returns (address) {
        address owner = _owners[tokenId];
        require(owner != address(0), "ERC721: owner query for nonexistent token");
        return owner;
    }

    /**
     * @dev See {IERC721Metadata-name}.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev See {IERC721Metadata-symbol}.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev See {IERC721Metadata-tokenURI}.
     */
    function tokenURI(uint256 tokenId) public view virtual override returns (string memory) {
        require(_exists(tokenId), "ERC721Metadata: URI query for nonexistent token");

        string memory baseURI = _baseURI();
        return bytes(baseURI).length > 0 ? string(abi.encodePacked(baseURI, tokenId.toString())) : "";
    }

    /**
     * @dev Base URI for computing {tokenURI}. If set, the resulting URI for each
     * token will be the concatenation of the `baseURI` and the `tokenId`. Empty
     * by default, can be overriden in child contracts.
     */
    function _baseURI() internal view virtual returns (string memory) {
        return "";
    }

    /**
     * @dev See {IERC721-approve}.
     */
    function approve(address to, uint256 tokenId) public virtual override {
        address owner = ERC721.ownerOf(tokenId);
        require(to != owner, "ERC721: approval to current owner");

        require(
            _msgSender() == owner || isApprovedForAll(owner, _msgSender()),
            "ERC721: approve caller is not owner nor approved for all"
        );

        _approve(to, tokenId);
    }

    /**
     * @dev See {IERC721-getApproved}.
     */
    function getApproved(uint256 tokenId) public view virtual override returns (address) {
        require(_exists(tokenId), "ERC721: approved query for nonexistent token");

        return _tokenApprovals[tokenId];
    }

    /**
     * @dev See {IERC721-setApprovalForAll}.
     */
    function setApprovalForAll(address operator, bool approved) public virtual override {
        _setApprovalForAll(_msgSender(), operator, approved);
    }

    /**
     * @dev See {IERC721-isApprovedForAll}.
     */
    function isApprovedForAll(address owner, address operator) public view virtual override returns (bool) {
        return _operatorApprovals[owner][operator];
    }

    /**
     * @dev See {IERC721-transferFrom}.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        //solhint-disable-next-line max-line-length
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: transfer caller is not owner nor approved");

        _transfer(from, to, tokenId);
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        safeTransferFrom(from, to, tokenId, "");
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) public virtual override {
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: transfer caller is not owner nor approved");
        _safeTransfer(from, to, tokenId, _data);
    }

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * `_data` is additional data, it has no specified format and it is sent in call to `to`.
     *
     * This internal function is equivalent to {safeTransferFrom}, and can be used to e.g.
     * implement alternative mechanisms to perform token transfer, such as signature-based.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeTransfer(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) internal virtual {
        _transfer(from, to, tokenId);
        require(_checkOnERC721Received(from, to, tokenId, _data), "ERC721: transfer to non ERC721Receiver implementer");
    }

    /**
     * @dev Returns whether `tokenId` exists.
     *
     * Tokens can be managed by their owner or approved accounts via {approve} or {setApprovalForAll}.
     *
     * Tokens start existing when they are minted (`_mint`),
     * and stop existing when they are burned (`_burn`).
     */
    function _exists(uint256 tokenId) internal view virtual returns (bool) {
        return _owners[tokenId] != address(0);
    }

    /**
     * @dev Returns whether `spender` is allowed to manage `tokenId`.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function _isApprovedOrOwner(address spender, uint256 tokenId) internal view virtual returns (bool) {
        require(_exists(tokenId), "ERC721: operator query for nonexistent token");
        address owner = ERC721.ownerOf(tokenId);
        return (spender == owner || getApproved(tokenId) == spender || isApprovedForAll(owner, spender));
    }

    /**
     * @dev Safely mints `tokenId` and transfers it to `to`.
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeMint(address to, uint256 tokenId) internal virtual {
        _safeMint(to, tokenId, "");
    }

    /**
     * @dev Same as {xref-ERC721-_safeMint-address-uint256-}[`_safeMint`], with an additional `data` parameter which is
     * forwarded in {IERC721Receiver-onERC721Received} to contract recipients.
     */
    function _safeMint(
        address to,
        uint256 tokenId,
        bytes memory _data
    ) internal virtual {
        _mint(to, tokenId);
        require(
            _checkOnERC721Received(address(0), to, tokenId, _data),
            "ERC721: transfer to non ERC721Receiver implementer"
        );
    }

    /**
     * @dev Mints `tokenId` and transfers it to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {_safeMint} whenever possible
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - `to` cannot be the zero address.
     *
     * Emits a {Transfer} event.
     */
    function _mint(address to, uint256 tokenId) internal virtual {
        require(to != address(0), "ERC721: mint to the zero address");
        require(!_exists(tokenId), "ERC721: token already minted");

        _beforeTokenTransfer(address(0), to, tokenId);

        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(address(0), to, tokenId);
    }

    /**
     * @dev Destroys `tokenId`.
     * The approval is cleared when the token is burned.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     *
     * Emits a {Transfer} event.
     */
    function _burn(uint256 tokenId) internal virtual {
        address owner = ERC721.ownerOf(tokenId);

        _beforeTokenTransfer(owner, address(0), tokenId);

        // Clear approvals
        _approve(address(0), tokenId);

        _balances[owner] -= 1;
        delete _owners[tokenId];

        emit Transfer(owner, address(0), tokenId);
    }

    /**
     * @dev Transfers `tokenId` from `from` to `to`.
     *  As opposed to {transferFrom}, this imposes no restrictions on msg.sender.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     *
     * Emits a {Transfer} event.
     */
    function _transfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {
        require(ERC721.ownerOf(tokenId) == from, "ERC721: transfer of token that is not own");
        require(to != address(0), "ERC721: transfer to the zero address");

        _beforeTokenTransfer(from, to, tokenId);

        // Clear approvals from the previous owner
        _approve(address(0), tokenId);

        _balances[from] -= 1;
        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(from, to, tokenId);
    }

    /**
     * @dev Approve `to` to operate on `tokenId`
     *
     * Emits a {Approval} event.
     */
    function _approve(address to, uint256 tokenId) internal virtual {
        _tokenApprovals[tokenId] = to;
        emit Approval(ERC721.ownerOf(tokenId), to, tokenId);
    }

    /**
     * @dev Approve `operator` to operate on all of `owner` tokens
     *
     * Emits a {ApprovalForAll} event.
     */
    function _setApprovalForAll(
        address owner,
        address operator,
        bool approved
    ) internal virtual {
        require(owner != operator, "ERC721: approve to caller");
        _operatorApprovals[owner][operator] = approved;
        emit ApprovalForAll(owner, operator, approved);
    }

    /**
     * @dev Internal function to invoke {IERC721Receiver-onERC721Received} on a target address.
     * The call is not executed if the target address is not a contract.
     *
     * @param from address representing the previous owner of the given token ID
     * @param to target address that will receive the tokens
     * @param tokenId uint256 ID of the token to be transferred
     * @param _data bytes optional data to send along with the call
     * @return bool whether the call correctly returned the expected magic value
     */
    function _checkOnERC721Received(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) private returns (bool) {
        if (to.isContract()) {
            try IERC721Receiver(to).onERC721Received(_msgSender(), from, tokenId, _data) returns (bytes4 retval) {
                return retval == IERC721Receiver.onERC721Received.selector;
            } catch (bytes memory reason) {
                if (reason.length == 0) {
                    revert("ERC721: transfer to non ERC721Receiver implementer");
                } else {
                    assembly {
                        revert(add(32, reason), mload(reason))
                    }
                }
            }
        } else {
            return true;
        }
    }

    /**
     * @dev Hook that is called before any token transfer. This includes minting
     * and burning.
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, ``from``'s `tokenId` will be
     * transferred to `to`.
     * - When `from` is zero, `tokenId` will be minted for `to`.
     * - When `to` is zero, ``from``'s `tokenId` will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {}
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0-rc.1 (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `recipient`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `sender` to `recipient` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address sender,
        address recipient,
        uint256 amount
    ) external returns (bool);

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}

// File: contracts/interfaces/INafta.sol



//SPDX-License-Identifier: Unlicense
pragma solidity ^0.8.9;

interface INafta {
  event AddNFT(
    address nftAddress,
    uint256 nftId,
    uint256 flashFee,
    uint256 pricePerBlock,
    uint256 maxLongtermBlocks,
    uint256 lenderNFTId,
    address msgSender
  );
  event EditNFT(
    address nftAddress,
    uint256 nftId,
    uint256 flashFee,
    uint256 pricePerBlock,
    uint256 maxLongtermBlocks,
    uint256 lenderNFTId,
    address msgSender
  );
  event RemoveNFT(address nftAddress, uint256 nftId, uint256 lenderNFTId, address msgSender);
  event WithdrawEarnings(uint256 earnings, address msgSender);
  event Flashloan(address nftAddress, uint256 nftId, uint256 earnedFees, address msgSender);
  event PoolFeeChanged(uint256 newPoolFee);
  event LongtermRent(address nftAddress, uint256 nftId, uint256 blocks, uint256 earnedFees, uint256 borrowerNFTId, address msgSender);

  // NFT Struct in the Pool (optimized to fit within 256 bits)
  struct PoolNFT {
    uint72 flashFee; // 72 bit - up to 4722.36648 ETH      // fee for a single rent in WETH,
    uint72 pricePerBlock; // 72 bit - up to 4722.36648 ETH      // price per block of longterm rent (0 if not available)
    uint24 maxLongtermBlocks; // 24 bit - up to 16 777 215 blocks   // maximum amount of blocks for longterm rent (0 if not available)
    uint32 inLongtermTillBlock; // 32 bit - up to block 4 294 967 295 // This NFT is in longterm rent till this block
    uint32 borrowerNFTId; // 32 bit - up to ID 4 294 967 296    // ID of BorrowerNFT
    uint32 lenderNFTId; // 32 bit - up to ID 4 294 967 296    // ID of LenderNFT (subtracted 2**32 to fit)
  }

  // BIG NFT Struct in the Pool (all values converted to uin256 for the ease of use)
  struct BigPoolNFT {
    uint256 flashFee; // originally 72 bit - up to 4722.36648 ETH      // fee for a single rent in WETH,
    uint256 pricePerBlock; // originally 72 bit - up to 4722.36648 ETH      // price per block of longterm rent (0 if not available)
    uint256 maxLongtermBlocks; // originally 24 bit - up to 16 777 215 blocks   // maximum amount of blocks for longterm rent (0 if not available)
    uint256 inLongtermTillBlock; // originally 32 bit - up to block 4 294 967 295 // This NFT is in longterm rent till this block
    uint256 borrowerNFTId; // originally 32 bit - up to ID 4 294 967 296    // ID of BorrowerNFT
    uint256 lenderNFTId; // originally 32 bit - up to ID 4 294 967 296    // ID of LenderNFT (add 2**32 to separate from BorrowerNFT)
  }

  // Getters
  function earnings(address userAddress) external returns (uint256);

  function poolNFTs(address nftAddress, uint256 nftId) external view returns (PoolNFT memory nft);

  function poolFee() external view returns (uint256 poolFee);

  function poolFeeChangedAtBlock() external view returns (uint256 poolFeeChangedAtBlock);

  function proposedOwner() external view returns (address proposedOwner);

  function borrowerNFTCount() external view returns (uint256 borrowerNFTCount);

  function lenderNFTCount() external view returns (uint256 lenderNFTCount);

  ////////////////////
  // Pool functions //
  ////////////////////

  /// @dev Add your NFT to the pool
  //
  /// @param nftAddress - The address of NFT contract
  /// @param nftId - ID of the NFT token you want to add
  /// @param flashFee - The fee user has to pay for a single rent (in WETH)
  /// @param pricePerBlock - If renting longterm - this is the price per block (0 if not renting longterm)
  /// @param maxLongtermBlocks - Maximum amount of blocks for longterm rent
  function addNFT(
    address nftAddress,
    uint256 nftId,
    uint256 flashFee,
    uint256 pricePerBlock,
    uint256 maxLongtermBlocks
  ) external;

  /// @dev Edit your NFT prices
  //
  /// @param nftAddress - The address of NFT contract
  /// @param nftId - ID of the NFT token you have in the pool
  /// @param flashFee - The fee user has to pay for a single rent (in WETH)
  /// @param pricePerBlock - If renting longterm - this is the price per block (0 if not renting longterm)
  /// @param maxLongtermBlocks - Maximum amount of blocks for longterm rent
  function editNFT(
    address nftAddress,
    uint256 nftId,
    uint256 flashFee,
    uint256 pricePerBlock,
    uint256 maxLongtermBlocks
  ) external;

  /// @dev Remove your NFT from the pool with earnings
  //
  /// @param nftAddress - The address of NFT contract
  /// @param nftId - ID of the NFT token you want to remove
  function removeNFT(address nftAddress, uint256 nftId) external;

  /// @dev Withdraw your earnings
  function withdrawEarnings() external;

  /// @dev Execute a Flashloan of NFT
  //
  /// @param nftAddress - The address of NFT contract
  /// @param nftId - ID of the NFT token you want to flashloan
  /// @param maxLoanPrice - Price the user is willing to pay for the flashloan
  /// @param receiverAddress - the contract that will receive the NFT (has to implement INFTFlashLoanReceiver interface)
  /// @param data - calldata that will be passed to the receiver contract (optional)
  function flashloan(
    address nftAddress,
    uint256 nftId,
    uint256 maxLoanPrice,
    address receiverAddress,
    bytes calldata data
  ) external;

  ///////////////
  // Flashlong //
  ///////////////

  /// @dev Utility function to update the status of longterm rent
  /// @dev Is called in some functions as well as can be called by public
  /// @dev will burn the BorrowerNFT if the longterm rent is over
  //
  /// @param nftAddress - The address of NFT contract
  /// @param nftId - ID of the NFT token you want to update
  function updateLongRent(address nftAddress, uint256 nftId) external;

  /// @dev You can buy a longterm rent for any NFT and don't pay fees for each use, and nobody else will be able to lend it while your rent lasts
  //
  /// @param nftAddress - The address of NFT contract
  /// @param nftId - ID of the NFT token you want to rent
  /// @param maxPricePerBlock - Price the user is willing to pay per block for renting the NFT
  /// @param receiverAddress - Who will receive the longterm rent BorrowerNFT
  /// @param blocks - How many blocks you want to rent (price is calculated per-block)
  function longRent(
    address nftAddress,
    uint256 nftId,
    uint256 maxPricePerBlock,
    address receiverAddress,
    uint256 blocks
  ) external;

  /////////////////////
  // Admin Functions //
  /////////////////////

  /// @dev Change the pool fee (admin only)
  /// @dev Only admin should be able to do that
  //
  /// @param newPoolFee - The new pool fee value (percentage)
  function changePoolFee(uint256 newPoolFee) external;

  /// @dev Propose a new owner, who will be able to claim ownership over this contract
  /// @param newOwner New owner who will be able to claim ownership over this contract
  function proposeNewOwner(address newOwner) external;

  /// @dev Claims the ownership of the contract if msg.sender is proposedOwner
  function claimOwnership() external;
}

// File: contracts/interfaces/IFlashNFTReceiver.sol



//SPDX-License-Identifier: Unlicense
pragma solidity ^0.8.9;

import {IERC721Receiver} from "@openzeppelin/contracts/token/ERC721/IERC721Receiver.sol";

interface IFlashNFTReceiver is IERC721Receiver {
    function executeOperation(address nftAddress, uint256 nftId, uint256 feeInWeth, address msgSender, bytes calldata data) external returns (bool);
}
// File: @openzeppelin/contracts/token/ERC721/utils/ERC721Holder.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0-rc.1 (token/ERC721/utils/ERC721Holder.sol)

pragma solidity ^0.8.0;

import "../IERC721Receiver.sol";

/**
 * @dev Implementation of the {IERC721Receiver} interface.
 *
 * Accepts all token transfers.
 * Make sure the contract is able to use its token with {IERC721-safeTransferFrom}, {IERC721-approve} or {IERC721-setApprovalForAll}.
 */
contract ERC721Holder is IERC721Receiver {
    /**
     * @dev See {IERC721Receiver-onERC721Received}.
     *
     * Always returns `IERC721Receiver.onERC721Received.selector`.
     */
    function onERC721Received(
        address,
        address,
        uint256,
        bytes memory
    ) public virtual override returns (bytes4) {
        return this.onERC721Received.selector;
    }
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0-rc.1 (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0-rc.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

// File: @openzeppelin/contracts/token/ERC721/IERC721Receiver.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0-rc.1 (token/ERC721/IERC721Receiver.sol)

pragma solidity ^0.8.0;

/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721Receiver {
    /**
     * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
     * by `operator` from `from`, this function is called.
     *
     * It must return its Solidity selector to confirm the token transfer.
     * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
     *
     * The selector can be obtained in Solidity with `IERC721.onERC721Received.selector`.
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4);
}

// File: @openzeppelin/contracts/token/ERC721/extensions/IERC721Metadata.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0-rc.1 (token/ERC721/extensions/IERC721Metadata.sol)

pragma solidity ^0.8.0;

import "../IERC721.sol";

/**
 * @title ERC-721 Non-Fungible Token Standard, optional metadata extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Metadata is IERC721 {
    /**
     * @dev Returns the token collection name.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the token collection symbol.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the Uniform Resource Identifier (URI) for `tokenId` token.
     */
    function tokenURI(uint256 tokenId) external view returns (string memory);
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0-rc.1 (utils/Address.sol)

pragma solidity ^0.8.0;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        assembly {
            size := extcodesize(account)
        }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0-rc.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/utils/Strings.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0-rc.1 (utils/Strings.sol)

pragma solidity ^0.8.0;

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _HEX_SYMBOLS = "0123456789abcdef";

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        // Inspired by OraclizeAPI's implementation - MIT licence
        // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        if (value == 0) {
            return "0x00";
        }
        uint256 temp = value;
        uint256 length = 0;
        while (temp != 0) {
            length++;
            temp >>= 8;
        }
        return toHexString(value, length);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _HEX_SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }
}

// File: @openzeppelin/contracts/utils/introspection/ERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.0-rc.1 (utils/introspection/ERC165.sol)

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}